
第一课

期货趋势型策略开发
和股票突破信号系统编写

系统化构建量化交易体系：

模块1：动手开发期货和股票量化交易策略

内容介绍

课程介绍

常用的量化平台概述

交易系统的核心要素

趋势型策略原理

用TB开发商品期货趋势型策略

用Python接入Tushare数据源来编写股票突破信号系统

课程介绍

系统化构建量化交易体系

三个模块

 动手开发期货和股票量化交易策略

 交易系统核心要素及多平台下的实例编写

 搭建自己的股票回测及交易平台

 从零实现一个自主可控的模块化交易系统

 量化交易策略逻辑的深入讨论

 策略改进方法、实战要点、上实盘的细节

 从数据、模型、回测、模拟、实盘全过程的
量化系统学习

 利用多种第三方工具平台开发策略

 对接开源数据接口开发策略

 搭建自己的回测和交易平台

 掌握主观策略的系统性量化方法

 掌握策略修正优化的思路和方法

 获得量化交易中各种实战技巧

 获取课件、源码、量化交易系统数据库方案

预期收获

常用的量化平台概述

自己动手做量化的起点

一个量化策略的进阶之路

 本地（图表/后台交易）

 金字塔、MC、TB、WH、TS、MT4、…

 云端（SaaS/券商定制）

 聚宽、优矿、米筐、…

 SDK/量化API（+终端/Web UI）

 万得、东财Choice、掘金量化、…

 开源框架（基于Python）

 PyCTP、VNPY、QuickLib、Zipline、…

第三方量化平台分类和举例

 丰富的Python交易和数据API接口

 事件驱动引擎（vn.event）

 开发示例（vn.demo）

 交易平台（vn.trader）

 RPC模块（vn.rpc）

 官方网站（www.vnpy.org）/知乎专栏

 官方交流QQ群

VN.PY

交易系统的核心要素

量化交易策略的“完整性”

常见的交易策略类型

 市场：买卖什么？

 头寸规模：买卖多少？

 入市：什么时候买进？

 止损：什么时候放弃一个亏损的头寸？

 退出：什么时候退出一个盈利的头寸？

 战术：怎么买卖？

交易系统的核心要素

趋势型策略原理

以“海龟交易法则”为例

海龟交易系统就是一个

包含了上述所有要素的

”完整的”交易系统

 波动性N：N =（19*PDN+TR）/20

 PDN = 前一日的N值

 TR = 当日的真实波动幅度

 真实波幅 = Max（H-L, |H-PDC|, |PDC-L|）

 H = 当日最高价

 L = 当日最低价

 PDC = 前一日收盘价

 头寸规模单位（手数）
= 账户的 1% /（N * 每一最小交易单位）

头寸规模计算

 假设股票中国平安的N值（ATR）为5元

 账户的本金为100万元，那么账户的1%就是
10000元

 最小交易单位是1手股票，也就是100股

 那么对于中国平安这只股票，可以分配的头
寸规模单位是：

 10000/（5*100）= 20（手股票）

 即：给中国平安分配的头寸规模单位是20手

（举例）

入市策略

 系统1：以20日突破
为基础的短期系统

 如上次突破是赢利性
突破，那么当前的入
市信号将被忽略

 55日突破点保障信号

 系统2：以55日突破
为基础的长期系统

 在突破点建立1个单
位的头寸

 按N/2的价格间隔逐
步扩大头寸

 以上一份订单的实际
成交价为基准

 直到总头寸达到规模
上限

逐步建仓

止损

 任何一笔交易风险程
度不超过账户的2%

 价格变动上限是2N

 对于加仓情况，止损
点上浮

 系统1：

 10日反向突破退出

 系统2：

 20日反向突破退出

退出

用TB开发商品期货趋势型策略

动手实践，从策略思想到编程实现，逐步求精

 运行TB，登录，新建工作区

 加载交易品种：螺纹钢（指数），日K级别

 设置商品，增加样本（K线）数到3000

 设置公式应用，允许连续建仓，最多10次，
最大持仓限制增加到2000合约

 新建“用户应用”Formula1；文件菜单，公
式管理，找到Formula1，选中并执行编译

 插入公式应用，加载Formula1

 如下页图，开始代码编辑……

环境准备

 建仓信号

 突破（20日）前高：开多信号

 跌破（20日）前低：开空信号

 仓位计算（无持仓情况下）

 突破时建立一个初始的头寸单位

 退出（有持仓情况下）

 跌破（10日）前低时平多

 突破（10日）前高时平空

初步实现

序列变量平移取值

N = AvgTR; ?

N = AvgTR[1];

AvgTR = XAverage(TrueRange, ATRLength);
N = AvgTR[1];

当前“实时刷新”的K线

突破入场示意图

入场点代码解析

1

2

DonchianHi

DonchianHi

// 多头开仓
if (high > DonchianHi && TurtleUnits >= 1)

{

myEntryPrice = min(high,

DonchianHi+MinPoint);

myEntryPrice = IIF(myEntryPrice < Open,

Open,

myEntryPrice); // 应对跳空

Buy(TurtleUnits, myEntryPrice);

}

1

2

休息一下

5分钟后回来

Params
Numeric RiskRatio(1);  // 1% risk ratio
Numeric ATRLength(20); // ATR window size
Numeric boLength(20);  // breakout
Numeric teLength(10);  // trailing exit

Vars
Numeric N;
NumericSeries AvgTR;
Numeric TotalEquity;
Numeric TurtleUnits;
NumericSeries DonchianHi;
NumericSeries DonchianLo;
Numeric MinPoint;
Numeric ExitLowestPrice;
Numeric ExitHighestPrice;
Numeric myEntryPrice;
Numeric myExitPrice;

Begin
if (!CallAuctionFilter()) return;

MinPoint = MinMove * PriceScale;
AvgTR = XAverage(TrueRange, ATRLength);
N = AvgTR[1];
TotalEquity = Portfolio_CurrentCapital() + Portfolio_UsedMargin();
TurtleUnits = (TotalEquity*RiskRatio/100) / (N*ContractUnit()*BigPointValue());
TurtleUnits = IntPart(TurtleUnits);

【初步实现的（简化版）代码】

DonchianHi = HighestFC(High[1], boLength);
DonchianLo = LowestFC(Low[1], boLength);
// PlotNumeric("Entry_Hi", DonchianHi);
// PlotNumeric("Entry_Lo", DonchianLo);

ExitLowestPrice = LowestFC(Low[1], teLength);
ExitHighestPrice = HighestFC(High[1], teLength);

if (MarketPosition == 0)
{

// 多头开仓
if (high > DonchianHi && TurtleUnits >= 1)
{

myEntryPrice = min(high, DonchianHi+MinPoint);
myEntryPrice = IIF(myEntryPrice < Open, Open, myEntryPrice);  // 应对跳空情况
Buy(TurtleUnits, myEntryPrice);

}
// 空头开仓
if (Low < DonchianLo && TurtleUnits >= 1)
{

myEntryPrice = max(low, DonchianLo-MinPoint);
myEntryPrice = IIF(myEntryPrice > Open, Open, myEntryPrice);  // 应对跳空情况
SellShort(TurtleUnits, myEntryPrice);

}
}

if (MarketPosition == 1)
{

if (Low < ExitLowestPrice)
{

// 多头退出
myExitPrice = max(Low, ExitLowestPrice-MinPoint);
myExitPrice = IIF(myExitPrice > Open, Open, myExitPrice); // 应对跳空情况
Sell(0, myExitPrice);

}
else {

// 多头加仓逻辑... 多头止损逻辑...
}

}
else if (MarketPosition == -1)
{

if (High > ExitHighestPrice)
{

// 空头退出
myExitPrice = min(High, ExitHighestPrice+MinPoint);
myExitPrice = IIF(myExitPrice < Open, Open, myExitPrice); // 应对跳空情况
BuyToCover(0, myExitPrice);

}
else {

// 空头加仓逻辑... 空头止损逻辑...
}

}
End

【螺纹钢】总收益：263.10%，最大回撤：14.98%，胜率：51.76%，盈亏比：2.49

【简化版回测结果】

 增仓动作（有持仓情况下）

 多头：相对于上次入场价格，每上涨N/2就新增
一个单位的多头头寸，每根K线内可多次加仓

 空头：相对于上次入场价格，每下跌N/2就新增
一个单位的空头头寸，每根K线内可多次加仓

 退出（有持仓情况下）

 新增规则：相对于上次入场价格，如超过最大
允许的损失幅度（2*N），则清仓

 加仓所在的那根K线内，不进行止损

完整逻辑

 长周期突破点保障信号

 入市过滤条件

 前次突破失败与否的判别

 盈利性退出

 亏损性退出

其它细节

if (MarketPosition == 0 && \
(!LastProfitableTradeFilter or PreBreakoutFailure))

{
// ...

}

Params
Numeric RiskRatio(1);                   // % Risk Per N ( 0 - 100)
Numeric ATRLength(20);                  // 平均波动周期 ATR Length
Numeric boLength(20);                   // 短周期 BreakOut Length
Numeric fsLength(55);                   // 长周期 FailSafe Length
Numeric teLength(10);                   // 离市周期 Trailing Exit Length
Bool LastProfitableTradeFilter(True);   // 使用入市过滤条件

Vars
Numeric MinPoint;                       // 最小变动单位
NumericSeries AvgTR;                    // ATR
Numeric N;                              // N 值
Numeric TotalEquity;                    // 按最新收盘价计算出的总资产
Numeric TurtleUnits;                    // 交易单位
NumericSeries DonchianHi;               // 唐奇安通道上轨，延后1个Bar
NumericSeries DonchianLo;               // 唐奇安通道下轨，延后1个Bar
NumericSeries fsDonchianHi;             // 唐奇安通道上轨，延后1个Bar，长周期
NumericSeries fsDonchianLo;             // 唐奇安通道下轨，延后1个Bar，长周期
Numeric ExitHighestPrice;               // 离市时判断需要的N周期最高价
Numeric ExitLowestPrice;                // 离市时判断需要的N周期最低价
Numeric myEntryPrice;                   // 开仓价格
Numeric myExitPrice;                    // 平仓价格
Bool SendOrderThisBar(False);           // 当前Bar有过交易
NumericSeries preEntryPrice(0);         // 前一次开仓的价格
BoolSeries PreBreakoutFailure(false);   // 前一次突破是否失败

Begin
// 集合竞价过滤
if (!CallAuctionFilter()) return; 【TB自带示例（完整版）代码】

If(BarStatus == 0)
{

preEntryPrice = InvalidNumeric;
PreBreakoutFailure = false;

}

MinPoint = MinMove*PriceScale;
AvgTR = XAverage(TrueRange,ATRLength);
N = AvgTR[1];
TotalEquity = Portfolio_CurrentCapital() + Portfolio_UsedMargin();
TurtleUnits = (TotalEquity*RiskRatio/100) /(N * ContractUnit()*BigPointValue());
TurtleUnits = IntPart(TurtleUnits); // 对小数取整

DonchianHi = HighestFC(High[1],boLength);
DonchianLo = LowestFC(Low[1],boLength);

fsDonchianHi = HighestFC(High[1],fsLength);
fsDonchianLo = LowestFC(Low[1],fsLength);

ExitLowestPrice = LowestFC(Low[1],teLength);
ExitHighestPrice = HighestFC(High[1],teLength);

Commentary("N="+Text(N));
Commentary("preEntryPrice="+Text(preEntryPrice));
Commentary("PreBreakoutFailure="+IIFString(PreBreakoutFailure,"True","False"));

// 当不使用过滤条件，或者使用过滤条件并且条件为PreBreakoutFailure为True进行后续操作
If(MarketPosition == 0 && ((!LastProfitableTradeFilter) Or (PreBreakoutFailure)))
{

// 突破开仓
If(High > DonchianHi && TurtleUnits >= 1)
{

// 开仓价格取突破上轨+一个价位和最高价之间的较小值，更接近真实情况，并能尽量保证成交
myEntryPrice = min(high,DonchianHi + MinPoint);
myEntryPrice = IIF(myEntryPrice < Open, Open,myEntryPrice); // 跳空用开盘价代替
preEntryPrice = myEntryPrice;
Buy(TurtleUnits,myEntryPrice);
SendOrderThisBar = True;
PreBreakoutFailure = False;

}

If(Low < DonchianLo && TurtleUnits >= 1)
{

// 开仓价格取突破下轨-一个价位和最低价之间的较大值，更接近真实情况，并能尽量保证成交
myEntryPrice = max(low,DonchianLo - MinPoint);
myEntryPrice = IIF(myEntryPrice > Open, Open,myEntryPrice); // 跳空用开盘价
preEntryPrice = myEntryPrice;
SendOrderThisBar = True;
SellShort(TurtleUnits,myEntryPrice);
SendOrderThisBar = True;
PreBreakoutFailure = False;

}
}

// 长周期突破开仓 Failsafe Breakout point
If(MarketPosition == 0)
{

Commentary("fsDonchianHi="+Text(fsDonchianHi));
If(High > fsDonchianHi && TurtleUnits >= 1)
{

// 开仓价格取突破上轨+一个价位和最高价之间的较小值，更接近真实情况，并能尽量保证成交
myEntryPrice = min(high,fsDonchianHi + MinPoint);
myEntryPrice = IIF(myEntryPrice < Open, Open,myEntryPrice); // 跳空用开盘价代替
preEntryPrice = myEntryPrice;
Buy(TurtleUnits,myEntryPrice);
SendOrderThisBar = True;
PreBreakoutFailure = False;

}

Commentary("fsDonchianLo="+Text(fsDonchianLo));
If(Low < fsDonchianLo && TurtleUnits >= 1)
{

// 开仓价格取突破下轨-一个价位和最低价之间的较大值，更接近真实情况，并能尽量保证成交
myEntryPrice = max(low,fsDonchianLo - MinPoint);
myEntryPrice = IIF(myEntryPrice > Open, Open,myEntryPrice); // 跳空用开盘价代替
preEntryPrice = myEntryPrice;
SellShort(TurtleUnits,myEntryPrice);
SendOrderThisBar = True;
PreBreakoutFailure = False;

}
}

If(MarketPosition == 1) // 有多仓的情况
{

Commentary("ExitLowestPrice="+Text(ExitLowestPrice));
If(Low < ExitLowestPrice)
{

myExitPrice = max(Low,ExitLowestPrice - MinPoint);
myExitPrice = IIF(myExitPrice > Open, Open,myExitPrice); // 跳空用开盘价代替
Sell(0,myExitPrice);    // 数量用0的情况下将全部平仓

}Else
{

If(preEntryPrice!=InvalidNumeric && TurtleUnits >= 1)
{

If(Open >= preEntryPrice + 0.5*N) // 如开盘超过设定的1/2N,则直接用开盘价增仓
{

myEntryPrice = Open;
preEntryPrice = myEntryPrice;
Buy(TurtleUnits,myEntryPrice);
SendOrderThisBar = True;

}

while(High >= preEntryPrice + 0.5*N) // 以最高价为标准，判断能进行几次增仓
{

myEntryPrice = preEntryPrice + 0.5 * N;
preEntryPrice = myEntryPrice;
Buy(TurtleUnits,myEntryPrice);
SendOrderThisBar = True;

}
}

// 止损指令
If(Low <= preEntryPrice - 2 * N && SendOrderThisBar == false) // 加仓Bar不止损
{

myExitPrice = preEntryPrice - 2 * N;
myExitPrice = IIF(myExitPrice > Open, Open,myExitPrice); // 跳空用开盘价代替
Sell(0,myExitPrice); // 数量用0的情况下将全部平仓
PreBreakoutFailure = True;

}
}

}Else If(MarketPosition ==-1) // 有空仓的情况
{

// 求出持空仓时离市的条件比较值
Commentary("ExitHighestPrice="+Text(ExitHighestPrice));
If(High > ExitHighestPrice)
{

myExitPrice = Min(High,ExitHighestPrice + MinPoint);
myExitPrice = IIF(myExitPrice < Open, Open,myExitPrice); // 跳空用开盘价代替
BuyToCover(0,myExitPrice);    // 数量用0的情况下将全部平仓

}Else
{

If(preEntryPrice!=InvalidNumeric && TurtleUnits >= 1)
{

If(Open <= preEntryPrice - 0.5*N) // 如开盘超过设定的1/2N,则直接用开盘价增仓
{

myEntryPrice = Open;
preEntryPrice = myEntryPrice;
SellShort(TurtleUnits,myEntryPrice);
SendOrderThisBar = True;

}

while(Low <= preEntryPrice - 0.5*N) // 以最低价为标准，判断能进行几次增仓
{

myEntryPrice = preEntryPrice - 0.5 * N;
preEntryPrice = myEntryPrice;
SellShort(TurtleUnits,myEntryPrice);
SendOrderThisBar = True;

}
}

// 止损指令
If(High >= preEntryPrice + 2 * N &&SendOrderThisBar==false) // 加仓Bar不止损
{

myExitPrice = preEntryPrice + 2 * N;
myExitPrice = IIF(myExitPrice < Open, Open,myExitPrice); // 跳空用开盘价代替
BuyToCover(0,myExitPrice); // 数量用0的情况下将全部平仓
PreBreakoutFailure = True;

}
}

}
End

【螺纹钢】总收益：700.22%，最大回撤：21.79%，胜率：50.09%，盈亏比：2.01

【完整版回测结果】

对比 -【简化版回测结果】

【螺纹钢】总收益：263.10%，最大回撤：14.98%，胜率：51.76%，盈亏比：2.49

 完整实现的效果一定比简化版好吗？

 思考问题的本质

 突破：趋势型策略，顺势加仓，在什么情况下
回调风险大？

 对品种、周期的选择有什么要求？

 尝试改写完整版的代码？

 同时进行跨市场、多品种的交易

 测试不同级别：60分钟、30分钟、周线

 测试多个品种：RB -> J，RU，PTA

思考

 移仓换月（主力合约切换）

 杠杆效应（保证金交易）

 波动幅度（隔夜跳空风险）

 套利机会（跨期、跨品种）

期货交易策略的注意事项

编写一个股票突破信号系统

用Python + Tushare免费数据源

 突破55日前高，作为买入信号

 细节：当日突破/前日未突破；日线复权处理

示例：突破信号

检测信号的一般代码流程

行情数据源：http://tushare.org

单只股票的信号检测

# -*- coding: utf-8 -*-

import tushare as ts

window_size = 5  # 过程演示用，待改为55
code = "000001"

temp = ts.get_k_data(code, start="2018-03-01", ktype="D", autype="qfq")
temp.index = temp.pop("date")
df = temp.loc[:, ["high", "close"]]
df["hhv"] = df["high"].rolling(window_size).max()
df["pre_hhv"] = df["hhv"].shift(1)
# df["signals"] = df["close"] > df["pre_hhv"]
df["signals"] = (df["close"].shift(1) <= df["pre_hhv"].shift(1)) & \

(df["close"] > df["pre_hhv"])
print(df)
results = df[df["signals"]]
print(list(results.index))

全市场股票的信号检测

# -*- coding: utf-8 -*-

import tushare as ts

window_size = 55

all_stocks = ts.get_stock_basics()["name"]
for code, name in all_stocks.iteritems():

temp = ts.get_k_data(code, start="2018-01-01", ktype="D", autype="qfq")
temp.index = temp.pop("date")
df = temp.loc[:, ["high", "close"]]
df["hhv"] = df["high"].rolling(window_size).max()
df["pre_hhv"] = df["hhv"].shift(1)
df["signals"] = (df["close"].shift(1) <= df["pre_hhv"].shift(1)) & \

(df["close"] > df["pre_hhv"])
results = df[df["signals"]]
if len(results) > 0:

print(code, name, list(results.index))

常用的第三方量化平台

完整的交易系统要素

从策略原理到编程实现的完整过程

平台·语言·数据·信号

总结

 用TB测试海龟交易法在不同周期、不同品种
上的表现，撰写测试报告

 测试不同级别：60分钟、30分钟、周线

 测试多个品种：RB -> J，RU，PTA

 对比完整版和简化版的表现，分析原因

 用Python和Tushare数据源，检测日线级别下
的“双均线交叉”信号

 上穿（金叉）或
下穿（死叉）的
动作如何定义？

课后练习

 题目：用Python开发均值回复型股票策略

 编程语言：Python

 运行平台：聚宽 - http://joinquant.com

 准备工作：

 提前在聚宽上注册账号

 阅读并熟悉聚宽的API文档

 自学Python和pandas

下节课预告

http://joinquant.com/

问答互动

在所报课的课程页面，
1、点击“全部问题”显示本课程所有学员提问的问题。
2、点击“提问”即可向该课程的老师和助教提问问题。

LOGO

小象学院：互联网新技术在线教育领航者

– 微信公众号：小象学院

联系我们

THANKS


